@ECHO OFF
CHCP 65001
COLOR 0A
TITLE PSHomeCacheDepot Cloner
SETLOCAL ENABLEEXTENSIONS DISABLEDELAYEDEXPANSION
SET "SRC1=%~DP0"
SET "GIT=C:\Program Files\Git\bin"
SET "GITEXE=%GIT%\git.exe"
SET "GITLFS1=C:\Program Files\Git\cmd\git-lfs.exe"
SET "GITLFS2=C:\Program Files\Git LFS\git-lfs.exe"
SET "REPOPATHS=%SRC1%folders.txt"
SET "ROOT=Repos"
SET "REPONAME=PSHomeCacheDepot"


:: CHECK IF GIT IS ALREADY IN THE PATH (OPTIONAL)
ECHO %PATH% | FIND /I "GIT\BIN" >NUL
IF %ERRORLEVEL%==0 (
    ECHO GIT IS ALREADY IN THE PATH.
    GOTO CONT
)

:: ADD GIT TO THE USER PATH
:: ADJUST THE PATH BELOW IF YOUR GIT INSTALLATION DIRECTORY IS DIFFERENT
SETX PATH "%PATH%;%GIT%"
ECHO GIT HAS BEEN ADDED TO YOUR ENVIRONMENT VARIABLES.

:CONT
IF NOT EXIST "%GIT%" ECHO GIT NOT FOUND @ END %GIT% & GOTO END

SETLOCAL ENABLEDELAYEDEXPANSION
SET "GITLFS="
IF NOT EXIST "%GITLFS1%" SET "GITLFS=!GITLFS!%GITLFS1%"
IF NOT "!GITLFS!" == "" (
    ECHO GIT LFS NOT FOUND @ !GITLFS!
    SET "GITLFSALT="
    IF NOT EXIST "%GITLFS2%" SET "GITLFSALT=!GITLFSALT!%GITLFS2%"
    IF NOT "!GITLFSALT!" == "" (
       ECHO GIT LFS NOT FOUND @ !GITLFS! OR !GITLFSALT!
    		GOTO END
    ) ELSE (
    	"%GITEXE%" lfs install
       GOTO MENU_MAIN
    )
) ELSE (
    "%GITEXE%" lfs install
    GOTO MENU_MAIN
)
ENDLOCAL

:MENU_MAIN
CD "%SRC1%"
ECHO.
ECHO.^|-------------^|
ECHO.^| GIT OPTIONS ^|
ECHO.^|-------------^|
ECHO.
ECHO 1. CLONE THE ENTIRE REPO
ECHO.
ECHO 2. CLONE SPECIFIC FOLDERS USING FOLDERS.TXT
ECHO.
ECHO 3. UPDATE LOCAL REPO USING GIT PULL
ECHO.
ECHO 4. SUSPEND CMD
ECHO.
SET /P CHOICE=ENTER A NUMBER TO PERFORM AN ACTION: 
ECHO.
IF NOT '%CHOICE%'=='' SET CHOICE=%CHOICE:~0,1%
IF '%CHOICE%'=='1' GOTO MAIN_CHOICE_ONE
IF '%CHOICE%'=='2' GOTO MAIN_CHOICE_TWO
IF '%CHOICE%'=='3' GOTO MAIN_CHOICE_THREE
IF '%CHOICE%'=='4' GOTO MAIN_CHOICE_FOUR

ECHO "%CHOICE%" IS NOT VALID, TRY AGAIN
ECHO.
GOTO MENU_MAIN

:MAIN_CHOICE_ONE
CALL :CLONE_ALL_FOLDERS
GOTO MAIN_OPTION_END

:MAIN_CHOICE_TWO
CALL :CLONE_SPECIFIC_FOLDERS
GOTO MAIN_OPTION_END

:MAIN_CHOICE_THREE
CALL :UPDATE_REPO
GOTO MAIN_OPTION_END

:MAIN_CHOICE_FOUR
IF EXIST "%SRC1%%REPONAME%" CD "%SRC1%%REPONAME%"
CMD /K
GOTO MAIN_OPTION_END

:MAIN_OPTION_END
CMD /K
GOTO MENU_MAIN

:END
CMD /K

::SSH- git clone git@hf.co:datasets/pebxcvi/PSHomeCacheDepot

:CLONE_ALL_FOLDERS

ECHO INIT FULL CLONE OF REPO

IF EXIST "%SRC1%%REPONAME%\.git" (
	CD "%SRC1%%REPONAME%"
	"%GITEXE%" sparse-checkout disable
	"%GITEXE%" pull
 ) ELSE (
 	"%GITEXE%" clone https://huggingface.co/datasets/pebxcvi/PSHomeCacheDepot
 )

GOTO MENU_MAIN

:CLONE_SPECIFIC_FOLDERS
ECHO INIT CUSTOM CLONE OF REPO
ECHO.
ECHO NOTE#1) THE INITIAL CREATION OF FOLDERS.TXT IS SET TO IGNORE ALL FOLDERS
ECHO REMOVE THE "#" FROM THE SELECTED FOLDERS THEN RERUN THE SCRIPT
ECHO.
ECHO NOTE#2) DELETING ENTRIES IN YOUR sparse-checkout FILE ...
ECHO WILL RESULT IN THE CORRESPONDING FOLDERS GETTING REMOVED FROM YOUR LOCAL REPO
ECHO.
ECHO NOTE#3) MUST HAVE MORE STORAGE AVAILABLE ON YOUR HDD THAN WHAT THE TOTAL REPO SIZE IS
ECHO.

IF NOT EXIST "%SRC1%%REPONAME%" (
	MKDIR "%SRC1%%REPONAME%"
	CD "%SRC1%%REPONAME%"
) ELSE (
	CD "%SRC1%%REPONAME%"
)

IF NOT EXIST "%SRC1%FOLDERS.TXT" (
	CALL :CREATE_TXT > "%SRC1%FOLDERS.TXT"
)

FOR /F "TOKENS=*" %%A IN ('DIR /B /A:-D "%SRC1%*FOLDERS*.TXT" 2^>NUL') DO SET "INPUT=%%A"

IF NOT EXIST "%SRC1%%REPONAME%\.git" (
	"%GITEXE%" init
	"%GITEXE%" remote add -f origin https://huggingface.co/datasets/pebxcvi/PSHomeCacheDepot
	"%GITEXE%" config core.sparseCheckout true
) ELSE (
  "%GITEXE%" config core.sparseCheckout true
   SETLOCAL ENABLEDELAYEDEXPANSION
   FOR /F "TOKENS=* USEBACKQ DELIMS=" %%A IN ("%SRC1%%INPUT%") DO (
     SET "A=%%A"
     SET "1ST=!A:~0,1!"
     SET "PTH=!A:#=!"
     SET "ACTPTH=!PTH:/=\!"
     IF EXIST "%SRC1%%REPONAME%\!ACTPTH!" (
     		IF /I "!PTH!"=="OBJECTS/" (
     			IF EXIST "%SRC1%%REPONAME%\OBJECTS\scee-home.playstation.net\c.home\prod2\live2\Objects\" (
     				IF EXIST "%SRC1%%REPONAME%\OBJECTS\scee-home.playstation.net\c.home\prod\live\Objects\" (
     					(ECHO !PTH!)>>"%SRC1%%REPONAME%\.git\info\sparse-checkout"
     				) ELSE (
     					BREAK
     				)
     			)
     		) ELSE IF /I "!PTH!"=="SCENES/" (
     			IF EXIST "%SRC1%%REPONAME%\SCENES\scee-home.playstation.net\c.home\prod2\live2\Scenes\" (
     				IF EXIST "%SRC1%%REPONAME%\SCENES\scee-home.playstation.net\c.home\prod\live\Scenes\" (
     					(ECHO !PTH!)>>"%SRC1%%REPONAME%\.git\info\sparse-checkout"
     				) ELSE (
     					BREAK
     				)
     			)
     		) ELSE (
     		 (ECHO !PTH!)>>"%SRC1%%REPONAME%\.git\info\sparse-checkout"
     		)
     )
   )
   ENDLOCAL  
   ECHO \.git ALREADY SET
)

SETLOCAL ENABLEDELAYEDEXPANSION
SET COUNTER=0
FOR /F "TOKENS=* USEBACKQ DELIMS=" %%A IN ("%SRC1%%INPUT%") DO (
    SET "A=%%A"
    SET "1ST=!A:~0,1!"
    IF NOT "!1ST!"=="#" (
        SET /A COUNTER+=1
        (ECHO !A!)>>"%SRC1%%REPONAME%\.git\info\sparse-checkout"
        ECHO !COUNTER! COPIED !A! TO .git\info\sparse-checkout
    )
)
ENDLOCAL

IF NOT EXIST "%SRC1%%REPONAME%\.git\info\sparse-checkout\" (
	IF EXIST "%SRC1%%REPONAME%\.git\info\sparse-checkout" (
		SORT /UNIQUE "%SRC1%%REPONAME%\.git\info\sparse-checkout" /O "%SRC1%%REPONAME%\.git\info\sparse-checkout"
	) ELSE (
		TYPE NUL > "%SRC1%%REPONAME%\.git\info\sparse-checkout"
	)
)

SETLOCAL ENABLEDELAYEDEXPANSION
IF EXIST "%SRC1%%REPONAME%\.git\info\sparse-checkout" IF NOT EXIST "%SRC1%%REPONAME%\.git\info\sparse-checkout\" (
	CALL :SETSIZE "%SRC1%%REPONAME%\.git\info\sparse-checkout"
	IF !SIZE! GTR 0 (
		"%GITEXE%" checkout main
	) ELSE (
	 	ECHO sparse-checkout IS EMPTY. UNCOMMENT AN ENTRY IN FOLDER.TXT.
	)
)
ENDLOCAL

GOTO MENU_MAIN

:UPDATE_REPO

ECHO INIT UPDATE OF REPO
ECHO.
ECHO.NOTE#4) IF ANY CHANGES ARE MADE TO YOUR LOCAL REPO, 
ECHO.SUSPEND CMD ( OPTION 4 ) THEN RUN "git stash" BEFORE YOU TRY TO PULL
ECHO. 

IF NOT EXIST "%SRC1%%REPONAME%" (
	ECHO NO GIT REPO FOUND @ "%SRC1%%REPONAME%"
) ELSE (
	CD "%SRC1%%REPONAME%"
	"%GITEXE%" pull
)

GOTO MENU_MAIN

:: EXAMPLE
:: Initialized empty Git repository in C:/Repos/PSHomeCacheDepot/.git/
:: Updating origin
:: remote: Enumerating objects: 635590, done.
:: remote: Counting objects: 100% (303481/303481), done.
:: remote: Compressing objects: 100% (284570/284570), done.
:: remote: Total 635590 (delta 11151), reused 30727 (delta 541), pack-reused 332109 (from 1)
:: Receiving objects: 100% (635590/635590), 640.75 MiB | 5.48 MiB/s, done.
:: Resolving deltas: 100% (99979/99979), done.
:: From https://huggingface.co/datasets/pebxcvi/PSHomeCacheDepot
::  * [new branch]            main       -> origin/main
:: 1 COPIED CONFIG XML/ TO .git\info\sparse-checkout
:: 2 COPIED MOTD&NEWSVIEWER/ TO .git\info\sparse-checkout
:: Updating index flags: 100% (579385/579385), done.
:: Updating files: 100% (516/516), done.
:: branch 'main' set up to track 'origin/main'.
:: Switched to a new branch 'main'

:SETSIZE
SET SIZE=%~z1
GOTO :EOF

:CREATE_TXT
@ECHO OFF
ECHO:# ADD # to ignore folder
ECHO:# REMOVE # to clone folder
ECHO:#CLUBHOUSE XML/
ECHO:#INVENTORY XML/
ECHO:#SCENE VERSION XMLS/
ECHO:#HELPTEXT/
ECHO:#COMMERCE POINTS/
ECHO:#RSS FEEDS/
ECHO:#SCREENS/
ECHO:#MOTD^&NEWSVIEWER/
ECHO:#SERVICE IDS/
ECHO:#OTHER XML/
ECHO:#CONFIG XML/
ECHO:#CONFIGS/
ECHO:#CONFIGS-MAPPED/
ECHO:#DEFAULTINVENTORY/
ECHO:#PROFANITYDICTIONARIES/
ECHO:#NAVIGATORS/
ECHO:#OBJECTCATALOGUES/
ECHO:#OBJECTCATALOGUES/OBJECTCATALOGUES-MIX-BETA/
ECHO:#OBJECTCATALOGUES/OBJECTCATALOGUES-MIX-DEC/
ECHO:#OBJECTCATALOGUES/OBJECTCATALOGUES-MIX-DEV/
ECHO:#OBJECTCATALOGUES/OBJECTCATALOGUES-MIX-ENC/
ECHO:#OBJECTS/
ECHO:#OBJECTS/scee-home.playstation.net/c.home/prod2/live2/Objects/
ECHO:#OBJECTS/scee-home.playstation.net/c.home/prod/live/Objects/
ECHO:#OBJECTS-MAPPED/
ECHO:#OBJECTS-MAPPED/ARCHIVE/
ECHO:#SCENELISTS/
ECHO:#SCENELISTS/DEC/
ECHO:#SCENELISTS/ENC/
ECHO:#SCENES/
ECHO:#SCENES/scee-home.playstation.net/c.home/prod2/live2/Scenes/
ECHO:#SCENES/scee-home.playstation.net/c.home/prod/live/Scenes/
ECHO:#SCENES-MAPPED/
ECHO:#SCENES-MAPPED/ARCHIVE/
ECHO:#ODC/
ECHO:#ODC/ODCsDecrypted/
ECHO:#SDC/
ECHO:#SDC/DecryptedSDCs/
ECHO:#DEV/
ECHO:#TSS/
ECHO:#AVATAR PFP/
ECHO:#MEDIA/
ECHO:#WEBSITE DUMPS/
ECHO:#MODDER DUMPS/
ECHO:#ALPHAZONE4/
EXIT /B